
Abstract 
This document provides conventions for QlikView script, specifically around file naming, file 

organization, indentation, comments, use of statements and use of white space. Additionally 

some general scripting recommendations are included.  

Introduction 
This document describes coding conventions for QlikView script and projects. This style guide is a 

living document, it evolves over time as additional conventions are identified and past conventions 

are rendered obsolete, either by changes in QlikView itself or as a result of new insights.  

 

Why you should use code conventions 

The most important reasons for having code conventions are readability and consistency. 

The Code Conventions for the Java™ Programming Language gives the following reasons for the 

importance of code conventions: 

“ 
• 80% of the lifetime cost of a piece of software goes to maintenance. 

• Hardly any software is maintained for its whole life by the original author. 

• Code conventions improve the readability of software, allowing developers 

to understand new code more quickly and thoroughly. 

 
” 

 

Know when to break the rules 

While consistency is good, there are cases where these code conventions just do not apply. Or as the 

Style Guide for Python  puts it: 

“ 
A style guide is about consistency. Consistency with this style guide is 

important. Consistency within a project is more important. Consistency within 

one module or function is most important. 

 

But most importantly: know when to be inconsistent -- sometimes the style 

guide just doesn't apply. When in doubt, use your best judgment. Look at other 

examples and decide what looks best. And don't hesitate to ask! 

 

In particular: do not break backwards compatibility just to comply with this PEP 

(ed: style guide)! 

 

Some other good reasons to ignore a particular guideline: 

 

• When applying the guideline would make the code less readable, even 

for someone who is used to reading code that follows this PEP. 

• To be consistent with surrounding code that also breaks it (maybe for 

historic reasons) -- although this is also an opportunity to clean up 

someone else's mess (in true XP style). 

• Because the code in question predates the introduction of the guideline 

and there is no other reason to be modifying that code. 

• When the code needs to remain compatible with older versions of 

Python that don't support the feature recommended by the style guide. 

 
” 

Conventions used in this document 

QlikView script is presented using the same formatting as the default QlikView script editor. Names 

are shown in Italic type, QlikView statements and functions used within sentences are shown in the 

Courier New  font. Emphasis is added by using bold type. 

Additionally, the following text boxes are used:  

� Identifies a tip or general good development practice. 

 

� Provides an additional, in-depth explanation of a subject that is mentioned in the 

text. 

 

� Identifies a warning or caveat, pay specific attention to these. 

 

“ Shows a quoted text, from a source other than those mentioned in the 

acknowledgements section. ” 
 

                                                           
1 http://www.oracle.com/technetwork/java/index-135089.html 
2 http://legacy.python.org/dev/peps/pep-0008/ 
3 http://williamlaygroup.com/ 

About Bitmetric 

Bitmetric is an independent consulting company that delivers Business 

Intelligence, Data Warehousing and Performance Management 

solutions. While we are able to deliver solutions in a wide range of 

technologies, we have a specific focus on QlikView. 

The company was founded in 2011 and is an organization that values technical and subject matter 

expertise. We are passionate about our profession and design and develop technical and 

organizational QlikView solutions, without losing sight of our client’s business realities. 

Sharing knowledge is important to us, not only with our clients but also with the wider QlikView 

community. We demonstrate this through our blogs (www.qlikfix.com),  books (QlikView 11 for 

Developers, Packt Publishing, November 2012, ISBN: 978-1849686068), training events 

(www.masterssummitforqlikview.com) and publications like this document. 

We offer both on-site and remote consultancy. If you are curious what we can do for you and your 

organization, then feel free to contact us using the contact details provided below. 

 

Contact details 

Office address Planetenweg 5 

2132 HN Hoofddorp 

The Netherlands 

 

Mailing address Postbus 79 

2130 AB Hoofddorp 

The Netherlands 

 

Telephone +31 23 230 2062 

 

Email info@bitmetric.nl 

 

Internet www.bitmetric.nl 

File names 
This section describes conventions used for naming files. 

 

File suffixes 

File suffixes must not be customized, rather, they should adhere to the standards of the respective 

applications. 

For example: 

• .qvw – QlikView application file 

• .qvd – QlikView data file 

• .qvx – QlikView data exchange file 

• .qvt – QlikView theme file 

• .qvs – QlikView script file 

 

File naming conventions 

Presentation, or “business”, applications must be named with a descriptive business term using a few 

words. This filename will appear exactly as typed in the QlikView AccessPoint web portal. 

QVD filenames will be precisely the name of the table it contains, plus the .qvd extension. Optionally, 

if a QVD contains merged/rolled-up data for a specific time period then a description of that period, 

separated by an underscore “_”, will be appended to the table name. 

QVD generators, i.e. the QVW workbook files that load, (optionally) transform data and store tables 

as QVDs will use all lower case, with underscore “_” delimiters. The phrase _qvd_gen and file 

extension .qvw will be appended to the end. 

QlikMarts, i.e. fully-associated data model QVW files without a user interface, will contain the name 

of the subject area in all lower case, with underscore “_” delimiters. If a specific filter or timeframe 

was applied to the QlikMart then this will be appended to the filename, separated with underscore 

“_” delimiters. The phrase _qlikmart and the file extension .qvw will be appended to the end. 

Some examples: 

• Sales Forecasting.qvw – business application, no prefix 

• Service Management.qvw – business application, no prefix 

• sales_fact_integration_qvd_gen.qvw 

• fiscal_calendar_qvd_gen.qvw 

• FiscalCalendar.qvd 

• InvoiceLine.qvd 

• InvoiceLine_2012.qvd 

• InvoiceLine_2013.qvd 

• InvoiceLine_YTD.qvd 

• inventory_analysis_qlikmart.qvw 

• sales_performance_ytd_qlikmart.qvw 

Included files 

Additional script statements can be inserted into a QlikView script using an “include” file. Typically, 

include files are used for: 

• Reusable script logic/subroutines 

• Configuration settings (e.g. locale or environment settings) 

• Data source connection strings 

• Shared/common variables 

• Globally-used calculations (saved as variables) 

Include filenames should be named in CamelCase and should have the .qvs file extension. CamelCase 

is defined by Wikipedia as: 

“ 
CamelCase (camel case) or medial capitals is the practice of writing compound 

words or phrases such that each word or abbreviation begins with a capital 

letter. Camel case may start with a capital or, especially in programming 

languages, with a lowercase letter. Common examples are PowerPoint or 

iPhone. 
” 

 

Some examples: 

• DataWarehouseConnection.qvs 

• SharedVariables.qvs 

• Config.qvs 

• CreateCalendar.qvs 

• LoadVariables.qvs 

 

� Please note that this document relates to scripting conventions only and does not 

prescribe any specific folder structure.  

File organization 
Each QlikView application file (.qvw) contains one script, typically organized across multiple tabs, and 

a front-end sheet (or sheets). 

 

QlikView application script layout and style 

The QlikView application script is a series of statements, definitions, etc. written in plain text in the 

QlikView scripting language. It is typically organized as a set of tabs – tabs are processed sequentially 

from left to right, top to bottom during loading. 

 

Script layout and style 

For legibility it is important to use a mono-spaced font. Courier New  will be used for all script 

text. This is the default setting in QlikView and must not be changed. Also, the default text 

highlighting for scripts will be used – this will automatically apply color and typeface settings to 

specific types of script statements: keywords, functions, etc. 

Also for legibility, limit the line length to 80 characters. This will require breaking long statements 

into multiple lines (this will be covered below). 

Information tab 

Each script must contain one Information tab – it is the first tab in the script. The Information tab will 

contain a comment block that includes: 

1. Name of the application 

2. Version number 

3. Author’s name – this is the original author of the application 

4. Date created 

5. Date last modified 

6. Brief description of the application 

Additionally, a change log will also be included on the Information tab. This includes: 

1. The version number 

2. Name of the developer that made the changes 

3. Modification date 

4. Brief summary of the changes 

An example is shown below (continued on next page): 

/************************************************** ******************* 
 
 Application:    Sales Dashboard.qvw 
 Version:     1.0 
 Author:     A. Developer 
 Date created:    2013/06/09 
 Date last modified:   2014/05/03 
 Description:     
  
 This application shows sales at the daily level fo r the past 3  
 years. Data is sourced from SAP and CRM, with addi tional targets  
 provided by Sales Management. 

--------------------------------------------------- ------------------- 
 
 Version:     1.0 
 Modified by:    An Other-Developer 
 Date modified:    2014/05/03 
  
 Changes: 
  
 - Change #157: switched source for CRM from SAP to  Microsoft  

  Dynamics CRM. 
 - Fixed bug #2506 (Targets incorrectly assigned to  months). 
 - Application approved for production. 
  
--------------------------------------------------- ------------------- 
 
 Version:     0.1 
 Modified by:    A. Developer 
 Date modified:    2013/06/09 
  
 Changes: 
  
 - Initial version. 
 
*************************************************** ******************/  
 

Initialize tab 

The Main tab that is created in QlikView script by default is renamed to Initialize and placed as the 

second tab in the script. This tab starts with the default SET statements that QlikView includes in 

each script to initialize system variables. For example: 

SET ThousandSep=','; 
SET DecimalSep='.'; 
SET MoneyThousandSep=','; 
SET MoneyDecimalSep='.'; 
SET MoneyFormat='$#,##0.00;($#,##0.00)'; 
SET TimeFormat='h:mm:ss TT'; 
SET DateFormat='M/D/YYYY'; 
SET TimestampFormat='M/D/YYYY h:mm:ss[.fff] TT'; 
SET MonthNames='Jan;Feb;Mar;Apr;May;Jun;Jul;Aug;Sep;Oct;Nov;Dec';  
SET DayNames='Mon;Tue;Wed;Thu;Fri;Sat;Sun';  

 

Please note that these values may appear slightly different depending on your localization settings. 

� 
The localization settings are a good candidate to be placed in an external QlikView script 

file (.qvs). This allows for easy reuse and updating of localization settings across multiple 

applications in the same environment. 

 

Next, the following optional settings are defined: 

1. Global environment settings – for example folder locations 

2. External scripts – code libraries etc. 

3. Constants – QlikView script has no real constant definition, so variables with the prefix const 

are used (more on this later). 

4. Variables – global variables used within the script or, optionally, overriding variables that 

were defined in external scripts (for example to overrule a specific environment-wide 

setting). 

An example is shown below: 

/*********************[ Global environment settings  ]*********************/  
$(must_include=..\00_include\00_config\config.qvs); 
 
 
/**************************[ External scripts ]**** ***********************/  
$(must_include=$(vFolder.Include.Scripts)createcalendarfromfield.qvs); 
$(must_include=$(vFolder.Include.Scripts)createcalendar.qvs); 
 
 
/******************************[ Constants ]******* ***********************/  
SET constYes = Dual('Yes', 1); 
SET constNo = Dual('No', 0); 
 
 
/******************************[ Variables ]******* ***********************/  
SET vScript.ExtractData = 'Y'; 
SET vScript.TransformData = 'Y';  

 

� 
The Dual function in QlikView allows for the assignment of different text and numeric 

values to a single field or variable. This makes flag fields, fields that indicate if a specific 

condition is met or not, a lot more useful. Calculations such as counts can use the 

numeric value (which yields better performance), while the text value can be shown as a 

filter value to the end-user. Unfortunately the numeric value cannot be easily referenced 

in Set Analysis as this compares to the string value by default. 

 

Subroutines tab 

Subroutines that are used across multiple QlikView applications should be placed in an external script 

file and included on the Initialize tab. Any subroutines that are local to the QlikView document (i.e. 

are only used in this specific document) should be placed in a tab named Subroutines. This tab is the 

third tab in the script. 

If there are many local subroutines the Subroutines tab may be split up into multiple tabs and 

grouped using grouping tabs (discussed below). 

 

EXIT tab 

Each script must contain an EXIT tab. This tab is used to identify the end of the script and is therefore 

typically the last tab in the script. The EXIT tab contains a single statement:  

EXIT SCRIPT;  

This statement tells QlikView to stop processing the script at this point. During development, the 

EXIT tab can be moved back and forth to easily limit the tabs that are processed during a reload. 

Script tabs and header comments 

Tabs provide a convenient mechanism to organize script content. Dedicate each tab to a major 

functional activity (e.g. loading a table, performing a calculation, performing a few related 

calculations). Use 80-100 lines of code per tab as a general rule-of-thumb. Restricting tabs to this size 

will improve code organization. 

Each tab must have a unique name (unique within the application). This will be useful when 

exporting a script to file, because all tabs are merged in that process. During the export process, a 

line will automatically be inserted at the top of each tab indicating the name: 

///$tab Information  
 

The name should be short but descriptive, preferably in the form verb subject. For example: 

• Extract SAP 

• Aggregate sales 

• Export tables 

Additionally, a header comment should be placed at the top of the tab (excluding the default tabs). 

This header should be kept simple an include only: 

1. The name of the tab 

2. A brief description of the purpose of the tab 

Example: 

/************************************************** ******************* 
 
 Tab: Extract SAP 
 Description:  
  
 This tab extracts (using an incremental load) the Campaign,  
 Adline, Account and Segment table from SAP and sto res it to QVD  
 files. 
 
*************************************************** ******************/  

 

Grouping tabs 

If multiple tabs belong to the same subject area, or perform similar operations, it might be 

worthwhile to add grouping tabs around them. These are simply empty tabs that contain, for the 

start tab, the name of the group followed by two greater-than signs (>>). The group is ended by 

another empty tab that contains the name of the group prefixed by two less-than signs (<<). 

Example: 

Indentation 
Four spaces should be used as the unit of indentation. The exact construction of indentation (spaces 

vs. tabs) is left to the developer’s choice. Space are more portable, but tabs require less keystrokes. 

 

Line length 

Avoid lines longer than 80 characters. Although some developers may have access to large and/or 

wide-screen monitors, plan for your scripts to be viewed on smaller displays. 

 

Wrapping lines 

When an expression will not fit on a single line, break it according to these general principles: 

• Break after comma 

• Break before an operator (+, -, etc.) 

• Prefer higher-level breaks to lower-level breaks (when parts of an expression are 

encapsulated in parentheses) 

• Align the new line with the beginning of the expression at the same level on the previous line 

• If the above rules lead to confusing code or code that is squished up against the right margin, 

just indent 4 spaces instead. 

Here are some example of breaking function calls: 

someFunction ( longExpression1 , longExpression2 , longExpression3 ,  
             longExpression4 , longExpression5 ); 
 
 
LET vResult = someFunction ( longExpression1 ,  
                           someOtherFunction ( longExpression2 ,  
                                             longExpression3 ));  
 
 
 

Following are two examples of breaking an arithmetic expression. The first is preferred, since the 

break occurs outside the parenthesized expression, which is at a higher level. 

 

 

Do: 
LET vResult = value1 * ( value2 + value3 + value4 -  value5 )  
              + ( value6 /  value7 ); 
 

Don’t: 
LET vResult = value1 * ( value2 + value3 + value4 
              -  value5 ) + ( value6 /  value7 ); 

 

Line wrapping for nested conditional should be carefully handled, because they can be difficult to 

interpret when evaluating code. The following page shows an example of line wrapping for nested 

conditionals: 

NestedConditional: 
LOAD 
 * 
RESIDENT 
 Temp_table 
WHERE 
ColA  = 1 
AND ColB  = 1 
OR 
( 
 ColA  = 2 
 AND ColB  = 3 
 AND 
 ( 
  ColC  = 0 
  OR ColC  = -1 
 ) 
) 
;  
 

Variable assignment 

It is suggested to not add tabs to align operators in variable assignments. 

Examples of operators: 

Do: 
SET vA = 2 + 3; 
SET vX = (2 + 3) / (4 * 5); 
SET vMyText = 'abc' & 'xyz'; 
 
 
 

Don’t 
SET vA      = 2 + 3; 
SET vX      = (2 + 3) / (4 * 5); 
SET vMyText = 'abc' & 'xyz'; 

Aliases 

Aligning aliases is recommended. While it does add some additional overhead during development it 

creates a tabular view that makes the script much more readable. One associated risk with aligning 

aliases is that non-aliased fields might be overlooked, especially when they precede or follow a multi-

line expression. Optionally, a developer may choose to align the non-aliased fields with the alias, or 

to create an alias with the same name (Facility AS Facility ). 

 

Do (continued on next page): 

// Load trouble ticket details from the Pink Elepha nt QVD 
TroubleTicketDetail: 
LOAD 
    Ticket       AS %TicketKey , 
    Ticket       AS FaultID , 
    [Event Type]      AS EventType , 
    date ( Date , 'MM/DD/YYYY')   AS ActualDate , 
    Facility , 
    Client       AS Contract , 
    Type , 
    Duration      AS [Duration (min)] , 
    1      AS TroubleTicketDetailCounter  
FROM 

    [Project QVD\extractJPE.qvd] ( qvd ) 
WHERE 
    Date  >= '12/01/2009'  
    AND Type  = 'Incident'  // logic from QOS report 
    AND Duration  > 0 
;   
 
 

Don’t: 

// Load trouble ticket details from the Pink Elepha nt QVD 
TroubleTicketDetail: 
LOAD 
    Ticket  AS %TicketKey , 
    Ticket  AS FaultID , 
    [Event Type]  AS EventType , 
    date ( Date , 'MM/DD/YYYY') AS ActualDate , 
    Facility , 
    Client  AS Contract , 
    Type , 
    Duration  AS [Duration (min)] , 
    1 AS TroubleTicketDetailCounter  
FROM 
    [Project QVD\extractJPE.qvd] ( qvd ) 
WHERE 
    Date  >= '12/01/2009'  
    AND Type  = 'Incident'  // logic from QOS report 
    AND Duration  > 0 
;   

 

� 
In order to get feedback on whether to indent aliases or not, a short survey was held 

under QlikView developers. The survey presented developers with 3 different options 

of formatting LOAD statements. Subsequently, it asked them which style they 

preferred to read and which style they were using in their own projects. 

 

Surprisingly, while over 80% preferred reading indented aliases, nearly half of 

respondents indicated that they were using the non-indented formatting in their own 

projects. 

 

The full survey can be found here:  

http://www.qlikfix.com/2014/06/16/aliases-indent/ 

Comments 
Comments should be used to give overviews of code and provide additional information that is not 

readily available in the code itself. Comments should contain only information that is relevant to 

reading and understanding the script. Discussion of nontrivial or non-obvious design decisions is 

appropriate, but avoid duplicating information that is present in (and clear from) the code. It is too 

easy for redundant comments to get out of date. In general, avoid any comments that are likely to 

get out of date as the script evolves. Comments that contradict the script are worse than no 

comments at all. 

Comments should be complete sentences. If a comment is a phrase or sentence, its first word should 

be capitalized, unless it is an identifier that begins with a lower case letter. Never alter the case of 

identifiers. 

If a comment is short, the period at the end can be omitted. Block comments generally consist of one 

or more paragraphs built out of complete sentences, and each sentence should end in a period. You 

should use two spaces after a sentence-ending period. 

Non-English QlikView developers should write their comments in English, unless they are 120% sure 

that code will never be read by people who do not speak their language. 

 

Block comments 

Block comments generally apply to some (or all) code that follows them, and are indented to the 

same level as that code. A block comment should be preceded by a blank line to set it apart from the 

rest of the code. Block comments start with a slash and an asterisk (/*) and end with an asterisk and 

a slash (*/).  

Here is an example of a block comment: 

/* 
This is a block comment. 
It has two lines. 
*/  

 

Single-line comments 

Short comments can appear on a single line indented to the level of script that follows. If a comment 

cannot be written in a single line, it should follow the block comment format (see previous section). 

A single-line comment uses two slashes (//) and should be preceded by a blank line. Here’s an 

example of a single-line comment: 

IF vScript.ExtractData  = 1 THEN 
 
 // Handle the condition. 
 LOAD * FROM file.csv; 
 
END IF  

 

Trailing comments 

Very short comments can appear on the same line as the script they describe, but should be shifted 

far enough to separate them from the statements. If more than one short comment appears in a 

chunk of code, they should all be indented to the same tab setting. Avoid the assembly language 

style of commenting every line of executable script with a trailing comment. Here’s an example of a 

trailing comment: 

IF vCondition  = 0 THEN 
 LOAD * FROM file.csv;  // Handle the condition. 
ELSE 
 LOAD * FROM file2.csv;  // This file must already exist. 
END IF  

 

End-of-line comments 

The // comment delimiter begins a comment that continues to the newline. It can comment out a 

complete line or only a partial line. It should not be used on consecutive multiple lines for text 

comments; however, it can be used in consecutive multiple lines for commenting-out section of code 

(see below for special instructions on commenting-out code). Examples of all three styles follow: 

IF vCondition  = 0 THEN 
 LOAD * FROM file.csv;  // Handle the condition 
ELSE 
 LOAD * FROM file2.csv;  // This alternate file must already 
exist 
END IF  
 
 
IF a = 1 THEN 
 
 // Load the CSV file first 
 LOAD * FROM file.csv; 
END IF  
 
 
// Due to intricacies of pos & neg values, a condit ional load must be used. 
IF x  > 0 THEN 
 LOAD * FROM pos.csv; 
ELSEIF x  < 0 
 LOAD * FROM neg.csv;  // Negative values are held in a 
separate file. 
ELSE 
 LOAD * FROM zero.txt;  // Zero is the default case. 
END IF  
 
// Remove the following comments before production use 
//MyTable: 
//LOAD * INLINE [ 
// X, Y 
// A, 1 
// B, 2 
//]; 
 
  

� 
Multiple script lines can be commented out in QlikView by selecting them, right-clicking 

and selecting Comment. To uncomment multiple lines, select them, right-click and select 

Uncomment. Alternatively, the keyboard shortcuts CTRL + K + C (comment) and CTRL + K 

+ U (uncomment) can be used. 

� 
End-of-line comments in INLINE tables 

Special care must be taken with INLINE tables and comments. No comments can 

be placed between the square brackets “[…]” of the table definition. Any text 

between those brackets will be interpreted as data. 

 

Wrong 
MyTable: 
LOAD * INLINE  [ 
 X, Y 
 A, 1   // Assign value 1 to item A. 
 B, 2 
 C, 3 
] ;  
 
This construction will actually include the comment as part of the Y-value for row A. 

Comments on inline tables should be placed outside of the square brackets. 

 

 

Commenting-out code 

Commenting-out lines (or partial lines) of script is a very effective technique when testing and 

debugging scripts. However, there should be absolutely no commented-out content in production 

script. Production script should be 100% clean to run and any ambiguities on “suspect” code should 

be resolved in the development and QA process. Commented-out code in the production script 

indicates poor design and implementation practice. 

Statements 
 

Simple statements 

Simple statements can, and should, occupy only a single line in the script. Typical examples of simple 

statements are: 

• Statements containing only a single keyword 

• Variable initialization 

• Trivial load or select statements that do not include field lists, or very short field lists 

Examples: 

DROP TABLE tmpMyTable; 
 
SET vYear = 2014; 
LET vDate = Today (); 
 
LOAD * RESIDENT Temp_ tmpMyTable; 
SQL SELECT customername FROM customer;  

 

LOAD statements 

Every script will include one or more LOAD statements. They are the mechanism for bringing data 

into the associative model. A LOAD statement will typically contain a number of keywords, such as: 

LOAD, SQL , SELECT, FROM, WHERE, GROUP BY, etc. 

These keywords should appear alone on their own line and each subsequent line is then indented by 

four spaces. Also, each field in the LOAD/SELECT list should be on its own line. Similarly for each 

condition in the WHERE clause. The closing semicolon is always placed on a separate line. This makes 

adding and removing WHERE, GROUP BY, ORDER BY clauses easier and introduces a natural 

whitespace that makes the script easier to read. 

Example: 

// Load trouble ticket details from the Pink Elepha nt QVD 
TroubleTicketDetail: 
LOAD 
    Ticket       AS %TicketKey , 
    Ticket       AS FaultID , 
    [Event Type]      AS EventType , 
    date ( Date , 'MM/DD/YYYY')   AS ActualDate , 
    Facility , 
    Client       AS Contract , 
    Type , 
    Duration      AS [Duration (min)] , 
    1      AS TroubleTicketDetailCounter  
FROM 
    [Project QVD\extractJPE.qvd] ( qvd ) 
WHERE 
    Date  >= '12/01/2009'  
    AND Type  = 'Incident'  // logic from QOS report 
    AND Duration  > 0 
;   

Exception for GROUP BY and ORDER BY 

An exception to the one-field-per-line rule can be made for simple GROUP BY and ORDER BY 

arguments. If there are multiple simple arguments, they can be placed on a single line after the 

keyword. 

Example: 

tmpCityPopulation: 
LOAD 
    City , 
    Country , 
    sum( NbrPpl ) AS Population  
RESIDENT 
    tmpPeopleByZipCode 
GROUP BY 
    City , Country  
ORDER BY 
    Country , City 
;  
 

Field order within LOAD statements 

There should always be a fixed order in which fields appear within a LOAD statement. Especially in 

larger tables, this makes it easier to find relevant fields: 

• Primary key: the key that uniquely identifies a single row within a table. Ideally this should be a 

single field, but can consist of multiple fields in the case of fact tables. 

• Foreign keys: keys that are used to associate a table with another table. Usually, this is the 

primary key of the table that is being linked to. 

• Attributes: descriptive labels/dates/classifications/etc. Fields that belong together should be 

grouped together. 

• Measures: fields that contain (semi-)additive facts. 

• Flag fields: indicator fields that specify if the row meets a certain condition (value 1) or not (value 

0).  Also see Flag fields in the Naming conventions section. 

• Counter fields: a field containing the value 1 for each row, which is used to easily count the 

number of rows within a selection. 

• Metadata fields: fields that contain information about the data. For example Valid from and 

Valid to fields used to describe the period that a record is valid in a Slowly Changing Dimension 

Type 2 scenario. Or, FactType to indicate from which fact table the row originates in a 

Concatenated Fact Table scenario. 

 

In a typical QlikView data model, not every table will contain all of these types of fields. Generally 

speaking Dimension tables will contain a Primary key, Attributes, Flag fields and Counter fields. Fact 

tables will contain Foreign keys, Measures and (optionally) Flag and Counter fields. Metadata fields 

are optional for both table types. 

� 
Dimension tables are tables that contain descriptions and attributes that provide 

context to measures stored in the Fact table. For example a Customer dimension 

table that contains attributes for each customer (name, account number, address, 

contact person, etc.). 

 

Fact tables contain measures. For example a Sales fact table that contains the 

number of products sold and their price. Fact tables link to their associated 

dimensions using a key field, for example %CustomerKey. 

 

Facts and Dimensions are used within the context of Dimensional Modeling. For 

those who are not familiar with this type of data modeling, the book The Data 

Warehouse Toolkit by Ralph Kimball and Margy Ross (Wiley, ISBN: 978-1118530801) 

offers a very good introduction to the subject. 

 

INLINE table loads 

Use the code formatted exactly as output from the Inline Wizard. Be sure to name the inline table 

and to include fieldnames. These fieldnames appear on the first row of the inline table. The closing 

semicolon can appear on the same line as the square closing bracket. 

Example: 

MyTable: 
LOAD * INLINE  [     
    Key, Value 
    A, 1 
    B, 2 
    C, 3 
] ;  
 

 

Include statements 

By default, QlikView will generate include statements in the following manner: 

 
$(include=createcalendar.qvs);  
 
 

During execution of the script, QlikView does not check if the included file actually exists. If the file 

does not exists the script will continue without generating any errors. Unless an include file is 

optional, include statements should always be rewritten to must_include ,  which does perform a 

check at runtime and will generate an error if the file does not exist. 

Example: 

$(must_include=createcalendar.qvs);   

IF..THEN..ELSEIF..END IF statements 

Each of the keywords: IF , ELSEIF , ELSE and END IF must begin on their own line. Immediately 

after the IF  or ELSEIF  keyword place the condition and end the line with the THEN keyword. 

Example: 

IF x  > 0 THEN 
 LOAD * RESIDENT tmpPos; 
ELSEIF x  < 0 THEN 
    LOAD * RESIDENT tmpNeg;   
ELSE 
    LOAD  
        A, 
        B, 
        sum( C) AS Summary 
    RESIDENT tmpZero 
    GROUP BY 
      A, B 
    ;     
END IF    

 

SWITCH..CASE..DEFAULT..END SWITCH statements 

Each of the keywords: SWITCH and END SWITCH must begin on their own line. The CASE keywords 

are placed on their own line, indented four spaces to the right of SWITCH. Similarly for the 

DEFAULT keyword. 

Example: 

SWITCH i  
    CASE 1 
        LOAD 1 AS case  AUTOGENERATE 1; 
    CASE 2 
        LOAD 2 AS case  AUTOGENERATE 1; 
    DEFAULT 
        LOAD 'Default' AS case  AUTOGENERATE 1; 
END SWITCH  

 

DO..EXIT DO..LOOP statements 

Each of the keyword: DO and LOOP must begin on their own line. Include the keyword WHILE or 

UNLESS and the condition statement on the same line as DO. 

If a break condition is included (using the EXIT  DO keyword), put it and its condition on its own line. 

It should be indented as any other statement within the loop. 

Example: 

LET a = 1; 
     
DO WHILE a < 10 
    LOAD * FROM [file$(a).csv]; 
    EXIT DO WHEN vSomeCondition = 0    // break out if this happens 
    LET a = a + 1; 
LOOP  

FOR..NEXT and FOR EACH..NEXT statements 

Each of the keywords: FOR, FOR EACH and NEXT must begin on their own line. Include the counter 

statement on the same line as FOR. If a STEP keyword is used to control the step size of the counter 

increments, include it on this line too. 

If a break condition is included (using the EXIT  FOR keyword), put it and its condition on its own 

line. It should be indented as any other statement within the loop. 

Examples: 

FOR a = 1 TO 9 
    LOAD * FROM [file$(a).csv];  
NEXT 
 
 
// more complex conditionals 
FOR counter = 1 TO 9 STEP 2 
    SET filename = x $(counter).csv; 
    IF rand () < 0.5 THEN 
        EXIT FOR UNLESS counter  = 1    // don’t break out on first 
iteration 
    END IF  
    LOAD * FROM [$(filename)]; 
NEXT  

 

� 
More flexible file handling 

Wildcards can be used within the QlikView LOAD statement to load several files at once. 

For example, the following statement will load the CustomerID and SalesAmount fields 

from every CSV file in the data folder: 

 
Data:  
LOAD 
 CustomerID , 
 SalesAmount  
FROM 
 [data\*.csv] 
;  
 

While this is a very useful feature, it does have its limitations when it comes to flexible 

processing of the files (for example, you cannot use a preceding load). The FOR EACH 
… NEXT statement offers an alternative that allows for greater flexibility: 

 
FOR EACH vFilename IN  fileList ('data\*.csv') 
  
 Data: 
 LOAD 
  CustomerID , 
  SalesAmount  
 FROM 
  [$(vFilename)] 
      ; 
 
NEXT  

White space 
 

Blank lines 

Blank lines improve readability by setting off sections of code that are logically related. 

Two blank lines should always be used in the following circumstances: 

• Between sections of a script file 

• After a LOAD statement, unless the next LOAD statement uses the CONCATENATE or JOIN  

prefix, in which case a single line is used. 

One blank line should always be used in the following circumstances: 

• Before a block or single-line comment (see Comments section above) 

• Between logical sections inside the script to improve readability 

 

Blank spaces 

Blank spaces should be used in the following circumstances: 

• A blank space should appear after commas in argument lists (see the Function Arguments 

section below) 

• A blank space should appear before and after an assignment operator, i.e. the equal sign ‘=’ 

• All binary operators should be separated from their operands by spaces. Blank spaces should 

never separate unary operators such as unary minus from their operands. 

 

� 
An operand is something on which an operation is performed, for example a field, 

value or variable. 

 

A unary operator is an operation that has only a single operand, for example: 

 
- $(vOne) 
 

This example takes the value of the variable vOne and changes the sign from positive 

to negative, or from negative to positive (if the value of vOne is already negative). 

 

A binary operator is an operation on two operands, for example: 

 

1 + 2 

 

In this example, both 1 and 2 are operands and + is a binary operator. 

 

Examples: 

SET vA = 2 + 3; 
SET vX = (2 + 3) / (4 * 5); 
SET vMyText = 'abc' & 'xyz'; 
SET vNegativeOne = - $(vOne);  // unary minus operation  

Naming conventions 
 

Tables 

Table names should be singular-tense nouns, in CamelCase with the first letter of each word 

capitalized. Try to keep your table names simple and descriptive. Use whole words – avoid acronyms 

and abbreviations, unless the abbreviation is much more widely used than the long form, such as URL 

or HTML. 

Examples: 

• Customer 

• OrderDetails 

• Invoice 

 

� Contrary to function and keyword names; table, field and variable names in 

QlikView are case-sensitive. 

 

Temporary tables 

Temporary tables are those that are created within a script to hold a temporary, intermediate table. 

They are manually dropped before the end of the script and not used in the final data model. 

Temporary tables are often invoked through the RESIDENT keyword. 

Temporary table names are to begin with the lower-case letters “tmp” and then a descriptive noun 

for the table in CamelCase (following the same approach as regular tables above). 

Examples: 

• tmpCustomer 

• tmpOrder 

• tmpInvoice 

• tmpSalesperson 

 

 

Mapping tables 

Mapping tables are special within QlikView, they always contain only two columns; a key and an 

associated lookup value. Mapping tables are automatically dropped at the end of the script. 

Mapping table names are to begin with the lower-case letters “map” and then a descriptive phrase 

for the table in CamelCase (following the same approach as regular tables above). This phrase should 

encompass the names of the key field and value field. 

Examples: 

• mapCityCountry 

• mapProductCategory 

• mapCustomerAccountmanager 

• mapOrderShipper 

Sorting tables 

Sorting tables are used to achieve a custom sort order on fields. Their names are to begin with the 

lower-case letters “sort”, followed by the name of the field that is being sorted. Sorting tables should 

always be dropped at the end of the script. 

Examples: 

• sortRegion 

• sortProductCategories 

• sortProduct 

• sortCustomerSegment 

 

� 
Sometimes there is a requirement that needs fields to be sorted in an order that is 

neither numeric nor alphabetical. Sorting tables are temporary tables that (ab)use a 

specific feature of QlikView, the Load Order.  

 

When data is loaded into QlikView, it is stored into Symbol tables4 in the order that it 

was first received. By creating a sorting table with the required sort order for the field 

before the actual data is loaded, we can ensure that QlikView stores the data in the 

order we specify. In the front-end, we can sort the field based on this custom sort 

order by selecting the Load Order option on the Sort tab. 

 

 
 

Sorting tables can be sourced from an external source (database, Excel, text file, etc.) 

or by using an inline table. The following example shows a custom sort order on a 

Region field: 

 
sortRegion: 
LOAD * INLINE  [     
    Region 
    North 
    South 
    East 
    West 
] ;  
 

 

 

  

                                                           
4 See Henric Cronström’s Symbol Tables and Bit-Stuffed Pointers blog post for a good introduction to how 

QlikView stores data: http://community.qlik.com/blogs/qlikviewdesignblog/2012/11/20/symbol-tables-and-

bit-stuffed-pointers 

Fields 

Field names should be short yet meaningful. The choice of a field name should be intuitive to the 

business user of the application, indicating the intent of its use. 

Because field names appear in the front-end visualizations (tables, charts, etc.) as-is, they should be 

coded as such in the script. In most cases it is possible to rename a field in the visualizations, but it 

will be more convenient (and consistent!) to name them appropriately in the script. 

Examples: 

• Fiscal period 

• Account type 

• Salesperson 

• Time to resolve 
 

 

Where the unit of measure is not abundantly obvious, it should be included in the field name. 

Examples: 

• Uptime (msec) 

• Revenue (m$) 

• Elapsed time (days) 

• Invoice age (days) 

 

To use multi-word field names in the script, they will need to be enclosed by square brackets […] . 

Examples: 

• [Fiscal period] 

• [Time to resolve (min)] 

 

 

� 
While renaming fields to user-friendly names is a good practice in user-facing 

applications, developers may want to use physical (the name of the field in the database) 

or shorter names in applications which are closer to the source system than to the end-

user. For example in QVD generators that extract and store historical data. The reasons 

for this are: 

 

• Source system experts are usually more familiar with fieldnames as they appear 

in the source system 

• Shorter names require less keystrokes 

 

Field names can be renamed in batch from physical names to user-friendly names by 

creating a mapping table that contains both of these names. At the end of the script, a 

mass renaming of the fields can then be performed using the following statement: 

 
RENAME FIELDS USING mapTable; 

Key fields 

Key fields are those fields that are intended to be “join” keys, i.e. associated with like-named fields in 

other tables. Key fields should never be used in the front-end application visualizations. 

To clearly highlight key fields, precede their name with a “%” percent sign and append the word 

“Key” on the end. Use CamelCase for key fields and do not separate words by blank spaces. 

Examples: 

• %DateKey 

• %CustomerKey 

• %InvoiceKey 

• %SalespersonKey 

 

To prevent key fields from accidentally being used by front-end designers or end-users; they can be 

automatically hidden by using the special variable HidePrefix. The syntax would be: 

SET HidePrefix = '%';  

 

� 
Keys are not data 

Key fields in a QlikView data model should only be used to map relationships 

between tables. Using keys in front-end visualizations can lead to wrong results, 

especially when using the count()  function. 

 

If the value of a key field, for example an order number, should be shown in the 

front-end, create a second, non-key field that contains that number. Additionally, if 

counting of records is required, it might be worthwhile to include a counter field 

containing only the value 1 (discussed below). 

 

More information and examples: 

http://qlikviewnotes.blogspot.nl/2013/09/autonumber-key-fields-and-

sequential.html 

 

 

Counter fields 

Counter fields are those fields that are used to count the number of records in a table. Typically 

these are defined as the literal value 1 for every row in the table (assuming the table grain is 

perfectly understood). Records in a table can then be easily counted by summing them. For example, 

to count the number of customers: 

sum( #CustomerCounter ) 

 

To clearly delineate counter fields, precede their name with a “#” pound sign and append the word 

“Counter” on the end. Between the # sign and the “Counter” label will be the table name. 

Examples: 

• #CustomerCounter 

• #OrderDetailCounter 

• #InvoiceCounter 

• #SalespersonCounter 

Flag fields 

Flag fields are similar to counter fields, except that they indicate if a certain condition (for example: 

was an invoice paid or not) is met. If the condition is met, the flag field is set to the value 1, if it is not 

met the flag field is set to the value 0. 

Using flag fields makes it much simpler to filter data in script and front-end application expressions.  

For example: 

Script: 

LOAD * RESIDENT tmpInvoice WHERE _InvoicePaid  = 1;  

Front-end: 

sum( [Sales amount] * _InvoicePaid ) 
sum({$< _CustomerActive = {1}, _OrderShipped = {1}>} [Order amount] )  
 
 

Flag fields are prefixed with an “_” underscore to distinguish them from other fields. The name of the 

field is the condition that has been tested in CamelCase. 

Examples: 

• _InvoicePaid 

• _CustomerActive 

• _OrderShipped 

• _CaseClosed 

 

� 
Set Analysis vs multiplication 

When writing front-end expressions, there are a few common approaches to limit 

aggregation of a measure based on a flag field. 

 

Multiplication: 

sum( [Sales amount]  * _InvoicePaid )  
 
Set Analysis: 

sum({$< _InvoicePaid = {1}>} [Sales amount] ) 
 
Experiments learn that there can be a dramatic difference in the calculation time 

between these two approaches. Though there really aren’t any hard-and-fast rules, 

the following default scenarios are suggested: 

 

• Measure in large fact table, flag in small dimension table: multiply 
• Measure in large fact table, flag in large dimension table: set analysis 
• Measure and flag both in same table: multiply 

 
Of course, the most important rule is: when in doubt, test.  

For completeness, if-statements that include a condition, such as: 

 
sum( if ( InvoiceStatus  = 'Paid', [Sales amount] , 0))   
 

should be avoided in front-end expressions at all times. 

Variables 

Variables are created in the QlikView script and in the front-end application (under Settings | 

Variable Overview). Variables names should be short yet meaningful. The choice of a variable should 

be mnemonic – that is, designed to indicate to the casual observer the intent of its use. 

Variable names start with a lower case prefix that indicates the type of contents the variable is 

holding, and then a descriptive noun for the variable in CamelCase: 

• Single values are prefixed with a “v”, for example vCurrentYear 

• Expressions are prefixed with an “e”, for example eSalesAmount 

• Constants, values that are never meant to be changed, are prefixed with “const”, for 

example constYes 

• Set Analysis statements are prefixed with an “s”, for example sYearToDate 

• Color expressions are prefixed with a “c”, for example cCaptionText 

• User defined functions are prefixed with an “f”, for example fCalcInterestRate 

One-character variables should be avoided except for temporary “throwaway” variables. Common 

names for temporary variables are i , j , k , m , n  for integers; x , y , z  for floats and c , d , e  for 

characters. Throwaway variables do not require a prefix. 

 

� 
Storing variables outside of QlikView 

Creating variables in QlikView script also enables the option of storing variable 

definitions outside of the QlikView script. For example in an Excel or CSV file, or in a 

database. This allows reuse of variable definitions across multiple applications. An 

example of how to achieve this can be found here:  

 

http://www.qlikfix.com/2011/09/21/storing-variables-outside-of-qlikview/ 

 

Grouping variables 

When multiple variables center around a common topic it is appropriate to prefix these variables 

with a common stem, separated from the variable name by a “.” dot. This will make it easier to locate 

related variables in the Variable Overview. 

Example: 

eSales.Amount 

eSales.Discount 

eSales.Tax 

eSales.ShippingCosts 

 

When variables share multiple levels, sub-stems and dots can be used. These should be ordered from 

the generic to the specific. For example, when using Set Analysis variables: 

sMonth.Current.ToDate – current month, up to the selected day of the month 

sMonth.Current.Total – total for the current month 

sMonth.PreviousYear.ToDate – current month in the previous year, up to selected day of the month 

sMonth.PreviousYear.Total – total for the current month in the previous year 

 

 

System variables 

QlikView provides a number of built-in system variables for use within scripts. These variables will be 

named in CamelCase. 

Common system variables: 

Number Handling Special Variables Error Handling Value Handling 
ThousandSep HidePrefix ErrorMode NullDisplay 
DateFormat HideSuffix ScriptError NullInterpret 
DayNames QvPath ScriptErrorDetails  NullValue 
MonthNames QvRoot ScriptErrorCount OtherSymbol 

 

 

Functions 

Function names 

Due to QlikView’s loose syntax, built-in functions can be called using any case mixture. For example, 

the following three calls are equivalent: 

DayStart ()  
daystart () 
daySTART()  
 
Not only are QlikView functions case-insensitive, but it also mixes case within its documentation and 

automation features. 

 

Use lower-case for all function names, unless the function name is a compound word, in which case 

the first letter of the subsequent words are to be capitalized (CamelCase). 

 

Some examples: 

 
dayStart () 
day () 
convertToLocalTime ()  
 
 

Function arguments 

Function arguments are contained in parentheses. Leave no space between the function name and 

the opening parenthesis: 

Do: 
dayStart (…) 
year (…) 
applyMap (…)  

Don’t 
if (…) 
today (…) 
sum (…)  

 

Leave no space after/before the opening/closing parenthesis: 

Do: 
dayStart ('2010-01-01 12:00:00')  
year ( Date ) 
applyMap ('mapCustomer', cust_id )  

Don’t 
dayStart ( '2010-01-01 12:00:00' ) 
year ( Date ) 
applyMap ( 'mapCustomer', cust_id )  

 

If a function takes multiple arguments, separate them by a comma and a single space: 

Do: 
applyMap ('mapCustomer', cust_id ) 
date ( OrderDate , 'YYYY-MM-DD')  
if ( year  = 2014, 2014, 'other year')  

Don’t 
applyMap ('mapCustomer', cust_id )  
date ( OrderDate ,'YYYY-MM-DD')  
if ( year =2014,2014,'other year')  
 

 

Subroutines 

Subroutines should be named in the form of verb subject, with the first character in lower case and 

subsequent words capitalized in CamelCase. 

Examples: 

• createCalendar 

• dropTables 

• createLinkTable 

Parameters used in subroutines should be prefixed with a “p” to indicate that they are local to the 

subroutine.  

Example: 

SUB loadIncrement( pTable, pKey, pQvdFile, pDropTable)  

 

 

Keywords 

Use UPPER CASE for all keywords. 

Some common keywords: 

LOAD AND DROP TABLE CONNECT JOIN IF..THEN..ELSE  
FROM OR QUALIFY SET KEEP SWITCH..CASE 
WHERE LIKE UNQUALIFY PEEK INNER FOR..NEXT 
SELECT RESIDENT MAPPING NOCONCATENATE LEFT INTERVALMATCH 
GROUP BY CONCATENATE MAP..USING AS RIGHT STORE..INTO 

 

The QlikView script editor will also highlight these keywords in a blue, boldfaced font. 

Scripting practices 
 

Be explicit 

The key principle behind the next few practices is to be explicit in your intentions. Explicitly stating 

what you intend for the result to be will greatly increase your script’s readability. 

 

Always name tables 

If no name is specified for a table, QlikView will use the name of the source table. While this is often 

the correct name, it is preferable to explicitly assign a name to any table that gets loaded into 

QlikView. If the table naming convention is used, this will mean that table names are consistent 

across the entire QlikView application. 

 

Explicitly name fields 

Avoid use of LOAD *  or SELECT * , which loads every field found in the source table, as much as 

possible. Specifically when loading from external files or databases. Changes in the underlying table 

might cause for more fields to be returned than was originally intended. 

Exceptions where LOAD *  might be valid is in the case of RESIDENT loads (where the table is 

created in the QlikView script and the developer can control its contents) and preceding loads. 

 

Explicitly CONCATENATE 

QlikView will automatically concatenate successively loaded tables if they have the same number of 

fields and the fields have the exact same names. Although this could be viewed as “handy”, it is 

better to explicitly use the keyword CONCATENATE. This will make your script easier to understand, 

especially when the concatenate spans over multiple tabs. 

 

Specify the target table for CONCATENATE, JOIN and KEEP 

When using the CONCATENATE, JOIN  or KEEP keywords, if no table is specified QlikView will 

automatically assume that the last table that was loaded before the keyword is the target table. Like 

with implicit concatenation, this could again be viewed as “handy”, but it is also a risk. If during 

subsequent updates to the script an additional LOAD statement gets placed directly above the 

CONCATENATE, JOIN  or KEEP keyword, QlikView will perform the operation on that table instead 

of the original, intended target table. 

Always specify a target table when using CONCATENATE, JOIN  or KEEP. 

Example (continues on next pages): 

Customer: 
LOAD 
 [Customer ID] , 
 [Customer name] , 
 [Customer phone]  

FROM 
 [Customers.csv] 
; 
 
LEFT JOIN (Customer) 
LOAD 
 [Customer ID] ,   // Key 
 [Customer segment]  
FROM 
 [CustomerSegments.csv] 
;  
 

Specify key fields when using JOIN and KEEP 

QlikView uses a natural join when joining two tables together using a JOIN  or KEEP statement. 

 

� 
A natural join is a type of join where the fields on which two tables are joined are not 

explicitly defined, instead they are implicitly determined. Any field that occurs in both 

tables will be used to join the two tables together. The resulting table contains only 

one column for each pair of equally named columns. 

 

Later changes to the script may cause the JOIN  or KEEP statement to take more columns into 

consideration then was initially intended. These errors can be hard to spot. Always put the intended 

keys at the top of the LOAD statement and add an end-of-line comment to indicate that these are 

the fields that should be joined on. 

Example: 

Order: 
LOAD 
 [Order ID] , 
 [Order year] , 
 [Order month] , 
 [Order amount]  
FROM 
 [Orders.csv] 
; 
 
LEFT JOIN (Order) 
LOAD 
 [Order year] ,    // Key 
 [Order month] ,    // Key 
 [Order quarter]  
FROM 
 [Calendar.csv] 
;  

� 
Although there are many cases where joining on multiple fields is valid, very often it can 

be simpler and cleaner to merge multiple keys into a single, concatenated key. In the 

example above, both Order year and Order month fields could be merged into a single 

Order period field. See the section on Keys (further below) for an example. 

 

Use parentheses liberally 

It is generally a good idea to use parentheses liberally in expressions involving mixed operators to 

avoid operator precedence problems. Even if the operator precedence seems clear to you, it might 

not be to others – you should not assume that other developers know precedence as well as you do. 

Examples: 

Do: 
IF (( a = b) AND ( c = d)) THEN ...  
LET x = ( a * b) + c - 1;  
 

Don’t 
IF ( a = b AND c = d) THEN ...  
LET x = a * b + c - 1;   

 

� 
Wrap variables containing expressions in parentheses 

If a variable contains an expression, it is a good idea to wrap it in parentheses. Otherwise, 

when multiple variables are used in an expression, operator precedence can cause the 

calculation to behave in an unexpected manner.  

 

For example, consider the following two variables: 

 
SET eNetSalesAmount = sum( TotalAmount ) - sum( ShippingCosts ); 
SET vTaxrate = 0.21;  
 
If these two variables are subsequently used in an expression that calculates the sales tax 

amount: 

 
$(eNetSalesAmount) * $(vTaxRate)  
 

Operator precedence will cause the expression to calculate the wrong result; the shipping 

costs are multiplied by the tax rate and the resulting number is subtracted from the total 

sales amount. Wrapping the expression in the variable in parentheses prevents this problem 

from occurring: 

 
SET eNetSalesAmount = ( sum( TotalAmount ) - sum( ShippingCosts )); 

 

Avoid QUALIFY 

The QUALIFY statement prevents unwanted associations in a QlikView data model by prefixing the 

name of the table to the field name in the form of “tablename.fieldname”. While this can be a useful 

tool to quickly load data into QlikView and explore its structure, it is bad form to use QUALIFY in 

production applications. Specifically, QUALIFIED tables have issues when joining or concatenating 

other tables to them. 

Clean up 

Manage temporary tables 

Always explicitly drop temporary tables as soon as they are no longer needed using a DROP TABLE 
statement. Failing to do so will consume unneeded memory during the script execution. 

 

Manage temporary variables 

Variables that are created in the script are automatically added to the front-end Variable Overview 

after completion of the script. Certain variables, especially throwaway variables, should only be used 

within the script and are not relevant in the front-end. These variables should be removed before the 

end of the script by assigning a null()  value to the variable: 

 

LET i = null ();  

Assigning a null()  value to a variable will only prevent that variable from appearing in the front-

end Variable Overview if it did not already exist there. If the variable already exists, it has to be 

removed first. 

 

Keys 

Separate concatenated keys 

When concatenating separate keys into a single, concatenated key, always use an “_” underscore 

separator between keys. 

Consider these two rows with different keys: 

ProductID ProductCategoryID 

25 1 

2 51 

 

If no separator is used, when concatenating the keys the result will be the same (251) for both keys. 

Using a separator prevents this. 

Example: 

ProductID  & '_' & ProductCategoryID  AS %ProductKey  
 
 

AutoNumber concatenated keys, after debugging 

Key fields that are used to associate tables in the QlikView data model should ideally contain 

sequential, integer numbers. The autoNumber()  function can be used to turn any field value into 

an integer number. Use the name of the key field in the second, optional counter argument to 

ensure that each key field contains sequential numbers. 

Example: 

autoNumber ( CustomerID , '%CustomerKey') AS %CustomerKey  

 

As the output of the autoNumber()  function is a meaningless number, it is advisable to only 

perform this transformation once the data model has been fully validated. Additionally it is strongly 

advisable to check the associations in the data model once this transformation has been performed. 

� 
AutoNumber is only consistent within a QlikView application 

For the same input, the autoNumber()  function will always return the same 

output. However, this is only consistent within a single QlikView application. 

 

When using the autoNumber()  function in different QlikView applications for the 

same values, it is very likely that the output will be different. Do not use 

autoNumber()  when data is extracted, transformed and loaded by separate 

QlikView applications. Or, only apply the function in the final, presentation 

application. 

 

 

Minimize RAM and CPU utilization 

 

Remove unused fields 

Removing unused fields can be tricky as some might be needed later. The Document Analyzer5 

QlikView application is a useful, free tool to (amongst other things) profile QlikView applications for 

any unused fields. This application automatically generates a series of DROP FIELD statements for 

fields that are not being used, these statements can be copied and pasted into a separate Cleanup 

tab at the end of the script. If a field is needed at a later stage the DROP FIELD statement for that 

field can simply be removed. 

� 
Verify before dropping fields 

While the Document Analyzer is a fantastic tool, it might in some cases erroneously 

flag some fields as unused (for example, when they are used in collaboration objects 

by users). Always perform a quick sanity check on the DROP FIELD statements 

before implementing them. 

 

Use counter fields 

Each dimension table (i.e. tables that contain descriptive labels, for example Customer, Product, 

Salesperson, etc.) should include a counter field. This is a numeric field that is set to the value 1 for 

each record. This will be useful in visualizations or other front-end objects that require a count of the 

number of records in the analysis. 

  

                                                           
5 available from: http://robwunderlich.com/ 

Example: 

Employee: 
LOAD 
 [Employee ID] , 
 [Employee name] , 
 [Employee position] , 
 1 AS #EmployeeCounter  
FROM 
 [Orders.csv] 
;  
 

� 
Count(DISTINCT …) is not always bad 

A very persistent myth, one that is even being taught in official QlikView trainings and 

shows up in official documentation, is that count(DISTINCT …)  is a slow, single-

threaded operation that should be avoided. It is one of the main reasons that counter 

fields are being used. 

 

In reality, this operation was already optimized in QlikView 9. In some cases, 

count(DISTINCT …) will outperform other methods, such as using counter fields.  

 

More information and examples: 

http://community.qlik.com/blogs/qlikviewdesignblog/2013/10/22/a-myth-about-

countdistinct 

 

 

Split high cardinality fields 

The amount of compression that QlikView can apply to your data is directly related to the number of 

unique values in each field (i.e. cardinality). Splitting fields that have high cardinality into multiple 

fields can greatly reduce data storage requirements. 

For example, consider a timestamp field which contains both a date and the time of day. In a single 

year this leads to: 

365 days x 24 hours x 60 minutes x 60 seconds = 31.536.000 different possibilities. 

Splitting this field up into separate Date and Time fields results in only 365 different possibilities for 

the Date field, and 86.400 different possibilities for the Time field. Or, a reduction of more than 

99,7%. 

Alternatively, if the time portion is not needed within the QlikView application, it can simply be 

removed from the timestamp by using the floor()  function. 

 

Remove number formatting from measures 

Sometimes, QlikView will store numbers as Dual values; one value containing the actual number (for 

example: 25) and one value containing a textual, formatted representation of that number (for 

example $25.00).  

While this is a great feature, it can also consume extra memory. As measures are almost never shown 

using their textual representation (because they are used in aggregation functions, and these 

functions use their own formatting), it is recommended to remove the textual part in order to reduce 

memory consumption. This is achieved by using the Floor()  function or the Num()  function 

without a format string. 

For example: 

Sales: 
LOAD 
 ProductID     AS %ProductKey , 
 CustomerID     AS %CustomerKey , 
 num( SalesAmount )   AS SalesAmount , 
 num( SalesPrice )   AS SalesPrice , 
 floor (1)     AS #OrderCounter  
FROM 
 [Sales.csv] 
;  
 

Move calculations to the script 

If possible, move calculations to the script. This means that instead of calculating every time the user 

makes a selection, a calculation is performed only once during reload. Obvious candidates to be 

moved to the script are expressions that calculate labels, for example: 

if ( AmountOutstanding  <= 0, '0. No amount outstanding', 
   if ( AmountOutstanding  <= 1000, '1. 0 - 1000', 
      if ( AmountOutstanding  <= 10000, '1. 1000 - 10.000', 
         '2. >10.000' 
         ) 
      ) 
    ) AS AmountOutstandingClass  

 

or 

[Sales amount]  * [Sales discount%]  AS [Sales discount amount]  
 

Turn on logging 

Always turn on logging for your QlikView documents, this saves the logged messages to an external 

log file (which uses the filename of the QlikView application and appends the .log extension) and 

makes debugging easier. Logging can be turned on by selecting Settings | Document Properties in the 

front-end of QlikView and ticking the Generate Logfile checkbox on the General tab. 

Use variables for folder and file locations 

Instead of hardcoding the filename in each QlikView LOAD statement, create variables that hold the 

pathnames. This ensures that paths are used consistently and can be easily updated in a single 

location. If stored in an external script file, these path definitions can also be used across multiple 

QlikView applications. 

Examples: 

vFolder.Source – the folder that stores the file-based source data (in subfolders) 

vFolder.QVD.Source – the folder that stores the original QVD data, without transformations 

vFolder.QVD.Business – the folder that stores the QVD’s to which business rules have been applied 

vFolder.Script – the folder that contains the external script files 

 

� 
Use relative or UNC paths 

When developing applications that use external file-based source data or other external 

files, keep in mind that the QlikView Server (or more accurately, the QlikView 

Distribution Service) should be able to access the files using the same paths. It is 

advisable to use relative or UNC paths instead of absolute paths. 

Sample script 
 

Information tab 
/************************************************** ******************* 
 
 Application:    Sales Dashboard.qvw 
 Version:     1.0 
 Author:     A. Developer 
 Date created:    2013/06/09 
 Date last modified:   2014/05/03 
 Description:     
  
 This application shows sales at the daily level fo r the past 3  
 years. Data is sourced from SAP and CRM, with addi tional targets  
 provided by Sales Management. 
         
--------------------------------------------------- ------------------- 
 
 Version:     1.0 
 Modified by:    An Other-Developer 
 Date modified:    2014/05/03 
  
 Changes: 
  
 - Change #157: switched source for CRM from SAP to  Microsoft  

  Dynamics CRM. 
 - Fixed bug #2506 (Targets incorrectly assigned to  months). 
 - Application approved for production. 
  
--------------------------------------------------- ------------------- 
 
 Version:     0.1 
 Modified by:    A. Developer 
 Date modified:    2013/06/09 
  
 Changes: 
  
 - Initial version. 
 
*************************************************** ******************/  

Initialize tab 
SET ThousandSep=','; 
SET DecimalSep='.'; 
SET MoneyThousandSep=','; 
SET MoneyDecimalSep='.'; 
SET MoneyFormat='$#,##0.00;($#,##0.00)'; 
SET TimeFormat='h:mm:ss TT'; 
SET DateFormat='M/D/YYYY'; 
SET TimestampFormat='M/D/YYYY h:mm:ss[.fff] TT'; 
SET MonthNames='Jan;Feb;Mar;Apr;May;Jun;Jul;Aug;Sep;Oct;Nov;Dec';  
SET DayNames='Mon;Tue;Wed;Thu;Fri;Sat;Sun';  

 

/*********************[ Global environment settings  ]*********************/  
$(must_include=..\00_include\00_config\config.qvs); 
 
 
/**************************[ External scripts ]**** ***********************/  
$(must_include=$(vFolder.Include.Scripts)createcalendarfromfield.qvs); 
$(must_include=$(vFolder.Include.Scripts)createcalendar.qvs); 
 
 
/******************************[ Constants ]******* ***********************/  
SET constYes = Dual('Yes', 1); 
SET constNo = Dual('No', 0); 
 
 
/******************************[ Variables ]******* ***********************/  
SET vScript.ExtractData = 'Y'; 
SET vScript.TransformData = 'Y';  

 


